
E02 – Curve and Surface Fitting

E02BBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

E02BBF evaluates a cubic spline from its B-spline representation.

2 Specification

SUBROUTINE E02BBF(NCAP7, LAMDA, C, X, S, IFAIL)
INTEGER NCAP7, IFAIL
real LAMDA(NCAP7), C(NCAP7), X, S

3 Description

This routine evaluates the cubic spline s(x) at a prescribed argument x from its augmented knot set
λi, for i = 1, 2, . . . , n + 7, (see E02BAF) and from the coefficients ci, for i = 1, 2, . . . , q in its B-spline
representation

s(x) =
q∑

i=1

ciNi(x).

Here q = n̄ + 3, where n̄ is the number of intervals of the spline, and Ni(x) denotes the normalised
B-spline of degree 3 defined upon the knots λi, λi+1, . . . , λi+4. The prescribed argument x must satisfy
λ4 ≤ x ≤ λn̄+4.

It is assumed that λj ≥ λj−1, for j = 2, 3, . . . , n̄+ 7, and λn̄+4 > λ4.

If x is a point at which 4 knots coincide, s(x) is discontinuous at x; in this case, S contains the value
defined as x is approached from the right.

The method employed is that of evaluation by taking convex combinations due to de Boor [4]. For further
details of the algorithm and its use see Cox [1] and [3].

It is expected that a common use of E02BBF will be the evaluation of the cubic spline approximations
produced by E02BAF. A generalization of E02BBF which also forms the derivative of s(x) is E02BCF.
E02BCF takes about 50% longer than E02BBF.

5 Parameters

1: NCAP7 — INTEGER Input

On entry: n̄+7, where n̄ is the number of intervals (one greater than the number of interior knots,
i.e., the knots strictly within the range λ4 to λn̄+4) over which the spline is defined.

Constraint: NCAP7 ≥ 8.

2: LAMDA(NCAP7) — real array Input

On entry: LAMDA(j) must be set to the value of the jth member of the complete set of knots, λj

for j = 1, 2, . . . , n̄+ 7.

Constraint: the LAMDA(j) must be in non-decreasing order with LAMDA(NCAP7 − 3) >
LAMDA(4).

3: C(NCAP7) — real array Input

On entry: the coefficient ci of the B-spline Ni(x), for i = 1, 2, . . . , n̄+3. The remaining elements of
the array are not used.

4: X — real Input

On entry: the argument x at which the cubic spline is to be evaluated.

Constraint: LAMDA(4) ≤ X ≤ LAMDA(NCAP7 − 3).

5: S — real Output

On exit: the value of the spline, s(x).

6: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

The argument X does not satisfy LAMDA(4) ≤ X ≤ LAMDA(NCAP7 − 3).

In this case the value of S is set arbitrarily to zero.

IFAIL = 2

NCAP7 < 8, i.e., the number of interior knots is negative.

7 Accuracy

The computed value of s(x) has negligible error in most practical situations. Specifically, this value has
an absolute error bounded in modulus by 18 × cmax×machine precision, where cmax is the largest
in modulus of cj , cj+1, cj+2 and cj+3, and j is an integer such that λj+3 ≤ x ≤ λj+4. If cj , cj+1, cj+2

and cj+3 are all of the same sign, then the computed value of s(x) has a relative error not exceeding
20×machine precision in modulus. For further details see Cox [2].

8 Further Comments

The time taken by the routine is approximately C× (1+0.1× log(n̄+7)) seconds, where C is a machine-
dependent constant.

Note. The routine does not test all the conditions on the knots given in the description of LAMDA in
Section 5, since to do this would result in a computation time approximately linear in n̄ + 7 instead of
log(n̄+ 7). All the conditions are tested in E02BAF, however.

9 Example

Evaluate at 9 equally-spaced points in the interval 1.0 ≤ x ≤ 9.0 the cubic spline with (augmented) knots
1.0, 1.0, 1.0, 1.0, 3.0, 6.0, 8.0, 9.0, 9.0, 9.0, 9.0 and normalised cubic B-spline coefficients 1.0, 2.0, 4.0,
7.0, 6.0, 4.0, 3.0.

The example program is written in a general form that will enable a cubic spline with n̄ intervals,
in its normalised cubic B-spline form, to be evaluated at m equally-spaced points in the interval
LAMDA(4) ≤ x ≤ LAMDA(n̄ + 4). The program is self-starting in that any number of data sets
may be supplied.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* E02BBF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NC7MAX
PARAMETER (NC7MAX=200)
INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)

* .. Local Scalars ..
real A, B, S, X
INTEGER IFAIL, J, M, NCAP, R

* .. Local Arrays ..
real C(NC7MAX), LAMDA(NC7MAX)

* .. External Subroutines ..
EXTERNAL E02BBF

* .. Intrinsic Functions ..
INTRINSIC real

* .. Executable Statements ..
WRITE (NOUT,*) ’E02BBF Example Program Results’

* Skip heading in data file
READ (NIN,*)

20 READ (NIN,*,END=80) M
IF (M.GT.0) THEN

READ (NIN,*) NCAP
IF (NCAP+7.LE.NC7MAX) THEN

READ (NIN,*) (LAMDA(J),J=1,NCAP+7)
READ (NIN,*) (C(J),J=1,NCAP+3)
A = LAMDA(4)
B = LAMDA(NCAP+4)
WRITE (NOUT,*)
WRITE (NOUT,*)

+ ’ J LAMDA(J) B-spline coefficient (J-2)’
WRITE (NOUT,*)
DO 40 J = 1, NCAP + 7

IF (J.LT.3 .OR. J.GT.NCAP+5) THEN
WRITE (NOUT,99999) J, LAMDA(J)

ELSE
WRITE (NOUT,99999) J, LAMDA(J), C(J-2)

END IF
40 CONTINUE

WRITE (NOUT,*)
WRITE (NOUT,*)

+ ’ R Argument Value of cubic spline’
WRITE (NOUT,*)

DO 60 R = 1, M
X = (real(M-R)*A+real(R-1)*B)/real(M-1)
IFAIL = 0

*
CALL E02BBF(NCAP+7,LAMDA,C,X,S,IFAIL)

*
WRITE (NOUT,99999) R, X, S

60 CONTINUE
GO TO 20

END IF
END IF

80 STOP
*
99999 FORMAT (1X,I3,F14.4,F21.4)

END

9.2 Program Data

E02BBF Example Program Data
9
4
1.00
1.00
1.00
1.00
3.00
6.00
8.00
9.00
9.00
9.00
9.00
1.00
2.00
4.00
7.00
6.00
4.00
3.00

9.3 Program Results

E02BBF Example Program Results

J LAMDA(J) B-spline coefficient (J-2)

1 1.0000
2 1.0000
3 1.0000 1.0000
4 1.0000 2.0000
5 3.0000 4.0000
6 6.0000 7.0000
7 8.0000 6.0000
8 9.0000 4.0000
9 9.0000 3.0000
10 9.0000
11 9.0000

R Argument Value of cubic spline

1 1.0000 1.0000
2 2.0000 2.3779
3 3.0000 3.6229
4 4.0000 4.8327
5 5.0000 5.8273
6 6.0000 6.3571
7 7.0000 6.1905
8 8.0000 5.1667
9 9.0000 3.0000

